(última atualização em 08/11/2008) **UNIVERSIDADE FEDERAL DO RIO GRANDE DO SUL**

INSTITUTO DE MATEMÁTICA
DEPARTAMENTO DE MATEMÁTICA PURA E APLICADA
Disciplina: MAT01168 -Matemática Aplicada -Semestre Letivo 2008/2
Professoras: Elisabeta Gallicchio e Irene Strauch
TERCEIRA ÁREA

SÉRIES DE FOURIER UTILIZANDO O MAPLE

A Forma Harmônica ou A Forma Compacta (em termos da amplitude e da fase) da Série de Fourier de uma função periódica f(t) é dada por

$$\textbf{f(t)=} A_0 + \sum A_n \cos\left(w_n t - \theta_n\right) \qquad \textbf{ou} \qquad \textbf{f(t)=} A_0 + \sum A_n \sin\left(w_n t + \phi_n\right),$$

onde

$A_n = \sqrt{a_n^2 + b_n^2}$ é a n-ésima aplitude e $\theta_n = arc\tan\left(\frac{b_n}{a_n}\right)$ é a n-ésima fase ou $\phi_n = arc\tan\left(\frac{a_n}{b_n}\right)$

Tensão Periódica: Fenômeno de Gibbs, Forma Compacta e Espectro de Amplitude

Considere a tensão periódica definida por f(t)=piecewise(t>=0 and t<T/4 ,1,T/4<t and t<T,0), sendo que f(t+2π)=f(t)

```
> restart:with(plots):with(plottools):plottools[point]:Digits:=
  3:assume(n,integer):interface(showassumed=0):#Para não exibir n~
Warning, the name changecoords has been redefined

Warning, the name arrow has been redefined
```

```
> T:=2*Pi;L:=T/2;
```

$$T := 2\,\pi \tag{1.1}$$

$$L := \pi$$

A freqüência fundamental é w_1=2π/T. Portanto, w_1=1. E, todas as outras freqüências serão múltiplos da fundamental w_n=2nπ/T , n=1,2,3..

```
> w[1]:=2*Pi/T; w[n]:=n*2*Pi/T;
```

$$w_1 := 1 \tag{1.2}$$

$$w_{n\sim} := n\sim$$

Dando entrada à tensão

```
> f:=t->piecewise(t>=0 and t<T/4 ,1,T/4<t and t<T,0);
```

$$f := t \to piecewise\left(0 \le t \textbf{ and } t < \frac{T}{4}, 1, \frac{T}{4} < t \textbf{ and } t < T, 0\right) \quad \textbf{(1.3)}$$

Traçar o gráfico da tensão f(t):

```
> plot(f(t),t=0..T,thickness=2,scaling=constrained,discont=true,
  titlefont=[COURIER,DEFAULT,14],labels=["t","f(t)"],labelfont=
  [COURIER,DEFAULT,10],axesfont=[COURIER,DEFAULT,10],title=`f(t)=1
  se 0<t<Pi/4,f(t)=0 se Pi/4<t<2Pi,f(t+2Pi)=f(t)`);
```



Gráfico do prolongamento periódico da função

```
> h:=proc(t)
                                    if t>0 and t<2*Pi then f(t)
  elif
   t>2*Pi and t<4*Pi then f(t-2*Pi) elif
                                                             t>-2*
  Pi and t<0 then f(t+2*Pi)elif
                              t>-4*Pi and t<-2*Pi then f(t+4*Pi)
  fi end:
> G:=plot(h,-4*Pi..4*Pi,thickness=2,titlefont=[COURIER,DEFAULT,12],
  labels=["t", "f(t)"],labelfont=[COURIER,DEFAULT,10],axesfont=
  [COURIER,DEFAULT,10],title=`f(t)=1 se 0<t<Pi/4,f(t)=0 se
  Pi/4<t<2Pi, T=2Pi)`): G;
```



Determinar os coeficientes a_n da série de Fourier, para n =1,2,3,...(Note que se trata de uma extensão em Série de Fourier sobre um intervalo qualquer: CASO II, com c=0, c+2L=2π).

```
> a[n]:=(1/L)*int(f(t)*cos(2*Pi*n/T*t),t=0..T);
```

$$a_{n\sim} := \frac{\sin\left(\frac{\pi n\sim}{2}\right)}{\pi n\sim} \quad (1.4)$$

Calcular o coeficiente a_0

```
> a[0]:= simplify((1/L)*int(f(t),t=0..T));
```

$$a_0 := \frac{1}{2} \quad (1.5)$$

A seguir, os coeficientes b_n, para n=1,2,3,...

```
> b[n]:=(1/L)*int(f(t)*sin(2*Pi*n/T*t),t=0..T);
```

$$b_{n\sim} := -\frac{\cos\left(\frac{\pi n\sim}{2}\right) - 1}{\pi n\sim} \quad (1.6)$$

```
> serie_tensao:=a[0]/2+Sum(a[n]*cos((w[n]*t))+b[n]*sin((w[n]*t)),
  'n'=1..infinity);
```

$$serie\_tensao := \frac{1}{4} + \sum_{n=1}^{\infty} \left(\frac{\sin\left(\frac{\pi n\sim}{2}\right) \cos(n\sim t)}{\pi n\sim} - \frac{\left(\cos\left(\frac{\pi n\sim}{2}\right) - 1\right) \sin(n\sim t)}{\pi n\sim} \right) \quad (1.7)$$

Aproximar a função f(t) com a série truncada até n=1..2 termos.

```
> f_ap2(t):=a[0]/2+ sum(a[n]*cos(2*Pi*n/T*t)+b[n]*sin(2*Pi*n/T*t),
  n=1..2);
```

$$f\_ap2(t) := \frac{1}{4} + \frac{\cos(t)}{\pi} + \frac{\sin(t)}{\pi} + \frac{\sin(2\,t)}{\pi} \quad (1.8)$$

Traçar os gráficos da tensão dada e da sua aproximação em série de Fourier truncada.

```
> plot({f(t),f_ap2(t)},t=0..T,discont=true,titlefont=[COURIER,
  DEFAULT,12],labels=["t", "f(t)"],labelfont=[COURIER,DEFAULT,10],
  axesfont=[COURIER,DEFAULT,10],title=` Tensão e sua Série de
  Fourier para n=2`, thickness=2, color=[blue,red]);
```



Aumentando o número de termos, obtém-se uma melhor aproximação. Por exemplo, para n=1..3 termos:

```
> f_ap3:=a[0]/2+ sum(a[n]*cos(2*Pi*n/T*t)+b[n]*sin(2*Pi*n/T*t),n=1.
  .3);
> plot({f(t),f_ap3},t=0..T,discont=true,thickness=2, color=[blue,
  red],titlefont=[COURIER,DEFAULT,12],labels=["t", "f(t)"],
  labelfont=[COURIER,DEFAULT,10],axesfont=[COURIER,DEFAULT,10],
  title=`Tensão e Série de Fourier truncada em n=3 `);
```

$$f\_ap3 := \frac{1}{4} + \frac{\cos(t)}{\pi} + \frac{\sin(t)}{\pi} + \frac{\sin(2\,t)}{\pi} - \frac{1}{3}\,\frac{\cos(3\,t)}{\pi} + \frac{1}{3}\,\frac{\sin(3\,t)}{\pi}$$



Com n=1..5 termos

```
> f_ap5:=a[0]/2+sum(a[n]*cos(2*Pi*n/T*t)+b[n]*sin(2*Pi*n/T*t), n=1.
  .5);
```

$$f\_ap5 := \frac{1}{4} + \frac{\cos(t)}{\pi} + \frac{\sin(t)}{\pi} + \frac{\sin(2\,t)}{\pi} - \frac{1}{3}\,\frac{\cos(3\,t)}{\pi} + \frac{1}{3}\,\frac{\sin(3\,t)}{\pi} + \frac{1}{5}\,\frac{\cos(5\,t)}{\pi} + \frac{1}{5}\,\frac{\sin(5\,t)}{\pi} \quad \textbf{(1.9)}$$

```
> plot({f(t),f_ap5},t=0..T,discont=true,thickness=2, color=[blue,
  red],titlefont=[COURIER,DEFAULT,12],labels=["t", "f(t)"],
  labelfont=[COURIER,DEFAULT,10],axesfont=[COURIER,DEFAULT,10],
  title=`Tensão e Série de Fourier truncada em n=5`);
```




```
> f_ap7:=a[0]/2+sum(a[n]*cos(2*Pi*n/T*t)+b[n]*sin(2*Pi*n/T*t), n=1.
  .7);
> plot({f(t),f_ap7},t=0..T,discont=true,thickness=2, color=[blue,
  red],titlefont=[COURIER,DEFAULT,12],labels=["t", "f(t)"],
```

```
labelfont=[COURIER,DEFAULT,10],axesfont=[COURIER,DEFAULT,10],
title=`Tensão e Série de Fourier truncada em n=7`);
```

$$f\_ap7 := \frac{1}{4} + \frac{\cos(t)}{\pi} + \frac{\sin(t)}{\pi} + \frac{\sin(2\,t)}{\pi} - \frac{1}{3}\,\frac{\cos(3\,t)}{\pi} + \frac{1}{3}\,\frac{\sin(3\,t)}{\pi} + \frac{1}{5}\,\frac{\cos(5\,t)}{\pi} + \frac{1}{5}\,\frac{\sin(5\,t)}{\pi} + \frac{1}{3}\,\frac{\sin(6\,t)}{\pi} - \frac{1}{7}\,\frac{\cos(7\,t)}{\pi} + \frac{1}{7}\,\frac{\sin(7\,t)}{\pi}$$




```
> f_ap10:=a[0]/2+sum(a[n]*cos(2*Pi*n/T*t)+b[n]*sin(2*Pi*n/T*t), n=
  1..10);
> plot({f(t),f_ap10},t=0..T,discont=true,thickness=2, color=[blue,
  red],titlefont=[COURIER,DEFAULT,12],labels=["t", "f(t)"],
  labelfont=[COURIER,DEFAULT,10],axesfont=[COURIER,DEFAULT,10],
  title=`Tensão e Série de Fourier truncada em n=10`);
```

$$f\_ap10 := \frac{1}{4} + \frac{\cos(t)}{\pi} + \frac{\sin(t)}{\pi} + \frac{\sin(2\,t)}{\pi} - \frac{1}{3}\,\frac{\cos(3\,t)}{\pi} + \frac{1}{3}\,\frac{\sin(3\,t)}{\pi} + \frac{1}{5}\,\frac{\cos(5\,t)}{\pi} + \frac{1}{5}\,\frac{\sin(5\,t)}{\pi} + \frac{1}{3}\,\frac{\sin(6\,t)}{\pi} - \frac{1}{7}\,\frac{\cos(7\,t)}{\pi} + \frac{1}{7}\,\frac{\sin(7\,t)}{\pi} + \frac{1}{9}\,\frac{\cos(9\,t)}{\pi} + \frac{1}{9}\,\frac{\sin(9\,t)}{\pi} + \frac{1}{5}\,\frac{\sin(10\,t)}{\pi}$$




```
> f_ap15:=a[0]/2+sum(a[n]*cos(2*Pi*n/T*t)+b[n]*sin(2*Pi*n/T*t), n=
  1..15);
> plot({f(t),f_ap15},t=0..T,discont=true,thickness=2, color=[blue,
  red],titlefont=[COURIER,DEFAULT,12],labels=["t", "f(t)"],
```

```
labelfont=[COURIER,DEFAULT,10],axesfont=[COURIER,DEFAULT,10],
title=`Tensão e Série de Fourier truncada em n=15`);
```

$$f\_ap15 := \frac{1}{4} + \frac{\cos(t)}{\pi} + \frac{\sin(t)}{\pi} + \frac{\sin(2\,t)}{\pi} - \frac{1}{3}\,\frac{\cos(3\,t)}{\pi} + \frac{1}{3}\,\frac{\sin(3\,t)}{\pi} + \frac{1}{5}\,\frac{\cos(5\,t)}{\pi} + \frac{1}{5}\,\frac{\sin(5\,t)}{\pi} + \frac{1}{3}\,\frac{\sin(6\,t)}{\pi} - \frac{1}{7}\,\frac{\cos(7\,t)}{\pi} + \frac{1}{7}\,\frac{\sin(7\,t)}{\pi} + \frac{1}{9}\,\frac{\cos(9\,t)}{\pi} + \frac{1}{9}\,\frac{\sin(9\,t)}{\pi} + \frac{1}{5}\,\frac{\sin(10\,t)}{\pi} - \frac{1}{11}\,\frac{\cos(11\,t)}{\pi} + \frac{1}{11}\,\frac{\sin(11\,t)}{\pi} + \frac{1}{13}\,\frac{\cos(13\,t)}{\pi} + \frac{1}{13}\,\frac{\sin(13\,t)}{\pi} + \frac{1}{7}\,\frac{\sin(14\,t)}{\pi} - \frac{1}{15}\,\frac{\cos(15\,t)}{\pi} + \frac{1}{15}\,\frac{\sin(15\,t)}{\pi}$$



Note que, por mais que se aumente o número de termos na série, os gráficos da tensão e de sua série de Fourier não coincidirão completamente. A série da f(t) converge para f(t), porém, nos pontos (to), onde a tensão é descontínua, a série converge para a média dos limites laterais no ponto, isto é, para $\frac{1}{2}$ [f(to+)+f(to-)]. Veja que os limites laterais em to=π/4 são f((to-)=1 e f(to+)=0 e a série converge, neste ponto, para o valor 0.5 Alí, o limite inferior da amplitude da oscilação é cerca de 9% do valor da tensão no ponto. **A menor amplitude aproximada da oscilação neste ponto é ?**

```
> f_ap20:=a[0]/2+sum(a[n]*cos(2*Pi*n/T*t)+b[n]*sin(2*Pi*n/T*t), n=
  1..20):
> plot({f(t),f_ap20},t=0..T,discont=true,thickness=2, color=[blue,
  red],titlefont=[COURIER,DEFAULT,12],labels=["t", "f(t)"],
  labelfont=[COURIER,DEFAULT,10],axesfont=[COURIER,DEFAULT,10],
  title=`Tensão e Série de Fourier truncada em n=20`);
```



Para visualizar o fenômeno de Gibbs, será traçado o gráfico da série e da tensão original. Agora, com n=1..50 termos e com mais detalhes (para vê-lo em escala clique sobre o gráfico e, após, em 1:1, na barra do comandos)

```
> f_ap50:=a[0]/2+sum(a[n]*cos(2*Pi*n/T*t)+b[n]*sin(2*Pi*n/T*t), n=
  1..50):
```

```
> menor_valor_osc:=evalf(9/100)*1;reta:=1+menor_valor_osc;reta1:=-
  menor_valor_osc;
```

$$menor\_valor\_osc := 0.0900 \tag{1.10}$$

$$reta := 1.09$$

$$reta1 := -0.0900$$

```
> plot({f(t),f_ap50,reta,reta1},t=0..T,discont=true,thickness=2,
  color=[blue,red,green,yellow], xtickmarks=[0,1.5708,6.2332],
  ytickmarks=[.5,0.90,1],titlefont=[COURIER,DEFAULT,12],labels=
  ["t", "f(t)"],labelfont=[COURIER,DEFAULT,10],axesfont=[COURIER,
  DEFAULT,10],title=`Fenômeno de Gibbs-Série de Fourier truncada em
  n=50`);
```



Forma Compacta: f(t)=$A_0 + \sum A_n \cos(w_n t - \theta_n)$ **ou** **f(t)=** $A_0 + \sum A_n \sin(w_n t + \phi_n)$

```
> A[0]:=a[0]/2;
```

$$A_0 := \frac{1}{4} \tag{1.11}$$

A n-ésima amplitude

```
> A[n]:=sqrt(a[n]^2+b[n]^2);
```

$$A_{n\sim} := \sqrt{\frac{\sin\left(\frac{\pi\, n\sim}{2}\right)^2}{\pi^2\, n\sim^2} + \frac{\left(\cos\left(\frac{\pi\, n\sim}{2}\right) - 1\right)^2}{\pi^2\, n\sim^2}} \quad \textbf{(1.12)}$$

A n-ésima fase

```
> fase[n]:=arctan(b[n]/a[n]);
```

$$fase_{n\sim} := -\arctan\left(\frac{\cos\left(\frac{\pi\, n\sim}{2}\right) - 1}{\sin\left(\frac{\pi\, n\sim}{2}\right)}\right) \quad \textbf{(1.13)}$$

```
> tensao_compac:=A[0]+Sum(A[n]*cos(n*w[1]*t-fase[n]),n=1..infinity)
  ;
```

$$tensao\_compac := \frac{1}{4} + \sum_{n\sim=1}^{\infty} \sqrt{\frac{\sin\left(\frac{\pi\, n\sim}{2}\right)^2}{\pi^2\, n\sim^2} + \frac{\left(\cos\left(\frac{\pi\, n\sim}{2}\right) - 1\right)^2}{\pi^2\, n\sim^2}}\, \cos\left(n\sim t + \arctan\left(\frac{\cos\left(\frac{\pi\, n\sim}{2}\right) - 1}{\sin\left(\frac{\pi\, n\sim}{2}\right)}\right)\right) \quad \textbf{(1.14)}$$

Cálculo da fase para n=1..6 e das tensões compactas para n=1..6

```
> tensao_c[n]:=(A[n]*cos(w[n]*t-fase[n]));
```

$$tensao\_c_{n\sim} := \sqrt{\frac{\sin\left(\frac{\pi\, n\sim}{2}\right)^2}{\pi^2\, n\sim^2} + \frac{\left(\cos\left(\frac{\pi\, n\sim}{2}\right) - 1\right)^2}{\pi^2\, n\sim^2}}\, \cos\left(n\sim t + \arctan\left(\frac{\cos\left(\frac{\pi\, n\sim}{2}\right) - 1}{\sin\left(\frac{\pi\, n\sim}{2}\right)}\right)\right) \quad \textbf{(1.15)}$$

```
> for i from 1 to 10 do theta[i]:=evalf(subs(n=i,fase[n]))
  :theta_graus[i]:=evalf(convert(theta[i],degrees)):t_c[i]:=evalf
  (subs(n=i,tensao_c[n]));od:
> for k from 1 to 10 do lprint(fase[k]=theta_graus[k],   fase[k]=
  theta[k]*`rad`, termo[k]=t_c[k]):od;
```

```
fase[1] = 44.8*degrees, fase[1] = .785*rad, termo[1] = .449*cos(t-.785)
fase[2] = 90.0*degrees, fase[2] = 1.57*rad, termo[2] = .318*cos(2.*t-1.57)
fase[3] = -44.8*degrees, fase[3] = -.785*rad, termo[3] = .150*cos(3.*t+.785)
fase[4] = 0., fase[4] = 0., termo[4] = .254e-3*cos(4.*t)
fase[5] = 44.8*degrees, fase[5] = .783*rad, termo[5] = .897e-1*cos(5.*t-.783)
fase[6] = 90.0*degrees, fase[6] = 1.57*rad, termo[6] = .106*cos(6.*t-1.57)
fase[7] = -44.8*degrees, fase[7] = -.783*rad, termo[7] = .640e-1*cos(7.*t+.783)
fase[8] = 1.70*degrees, fase[8] = .298e-1*rad, termo[8] = .133e-2*cos(8.*
t-.298e-1)
fase[9] = 43.9*degrees, fase[9] = .767*rad, termo[9] = .489e-1*cos(9.*t-.767)
fase[10] = 90.0*degrees, fase[10] = 1.57*rad, termo[10] = .636e-1*cos(10.*t-1.57)
```

A série da tensão com n=0..6 é

```
> tensao_compacta:=A[0]+sum(t_c[n],n=1..10);
```

$$tensao\_compacta := \frac{1}{4} + 0.449\cos(t - 0.785) + 0.318\cos(2.\,t - 1.57) + 0.150\cos(3.\,t + 0.785) + 0.000254\cos(4.\,t) + 0.0897\cos(5.\,t - 0.783) + 0.106\cos(6.\,t - 1.57) + 0.0640\cos(7.\,t + 0.783) + 0.00133\cos(8.\,t - 0.0298) + 0.0489\cos(9.\,t - 0.767) + 0.0636\cos(10.\,t - 1.57) \tag{1.16}$$

```
> Graf1:=plot(tensao_compacta, t=0..2*Pi,color=magenta,thickness=2,
  titlefont=[COURIER,DEFAULT,12],labels=["t", "f(t)"],labelfont=
  [COURIER,DEFAULT,10],axesfont=[COURIER,DEFAULT,10],title=`Tensão
  Harmônica(10 primeiros termos,1 período)`):Graf1;
```



Cálculo das amplitudes, para o gráfico no domínio freqüência correspondente à série da tensão :

```
> for i from 1 to 10 do A[i]:=evalf(subs(n=i,A[n])) :  w[i]:=subs
  (n=i,w[n]):od:
> for k from 1 to 10 do lprint(frequencia[k]=w[k],amplitude[k]=A[k]
  ):od;
frequencia[1] = 1, amplitude[1] = .449
frequencia[2] = 2, amplitude[2] = .318
frequencia[3] = 3, amplitude[3] = .150
frequencia[4] = 4, amplitude[4] = .254e-3
frequencia[5] = 5, amplitude[5] = .897e-1
```

```
frequencia[6] = 6, amplitude[6] = .106
frequencia[7] = 7, amplitude[7] = .640e-1
frequencia[8] = 8, amplitude[8] = .133e-2
frequencia[9] = 9, amplitude[9] = .489e-1
frequencia[10] = 10, amplitude[10] = .636e-1
```

Para o espectro da freqüência, vamos definir w[0]=0 para a amplitude A_0

```
> w[0]:=0:
```

```
>  for i from 0 to 10 do g[i]:=line([w[i],0],[w[i],A[i]], color=
  red, thickness=3):od:
```

```
> Graf2:=plots[display]({g[0],g[1],g[2],g[3],g[4],g[5],g[6],g[6],g
  [7],g[8],g[9],g[10]},xtickmarks=[0,1,2,3,4,5,6,7,8,9,10],
  titlefont=[COURIER,DEFAULT,12],labels=["w[n]","A[n]"],labelfont=
  [COURIER,DEFAULT,10],axesfont=[COURIER,DEFAULT,10],title=
  `Espectro de Amplitude da Tensão, n=10`,axes=frame):Graf2;
```



Observa-se que o termo dominante na série é o primeiro termo , pois $A_1 := 0.449$.

==